

9. Using Excel matrices functions to calculate partial autocorrelations

In order to use a more elegant way to calculate the partial autocorrelations we need to borrow some equations from stochastic modelling, as well as to dive into the matrix algebra.

Autocorrelations and partial autocorrelations are also used to define certain stochastic processes, and one of them is called autoregressive process (introduced in Chapter 11). For certain stochastic processes, the autocorrelation function satisfies the following equation:

$$\rho_1 = \phi_{k1}\rho_{j-1} + \dots + \phi_{k(k-1)}\rho_{j-k+1} + \phi_{kk}\rho_{j-k} \quad (9.1)$$

In equation (9.1), ρ_j are the autocorrelation coefficients that depend on some lagged value of the coefficients ρ_{j-k} , as well as on ϕ_{kk} , which are the partial autocorrelation coefficients.

Using the matrix notation, equation (9.1) can be represented as:

$$\begin{bmatrix} \rho_1 \\ \rho_2 \\ \cdot \\ \cdot \\ \rho_k \end{bmatrix} = \begin{bmatrix} 1 & \rho_1 & \rho_2 & \dots & \rho_{k-1} \\ \rho_2 & 1 & \rho_1 & \dots & \rho_{k-2} \\ \cdot & \cdot & \cdot & \cdot & \cdot \\ \cdot & \cdot & \cdot & \cdot & \cdot \\ \rho_{k-1} & \rho_{k-2} & \rho_{k-3} & \dots & 1 \end{bmatrix} \begin{bmatrix} \phi_{k1} \\ \phi_{k2} \\ \cdot \\ \cdot \\ \phi_{kk} \end{bmatrix} \quad (9.2)$$

The alternative expression for (9.2) is:

$$\boldsymbol{\rho}_k = \mathbf{P}_k \boldsymbol{\phi}_k \quad (9.3)$$

This means that:

$$\boldsymbol{\phi}_k = \frac{\boldsymbol{\rho}_k}{\mathbf{P}_k} = \mathbf{P}_k^{-1} \boldsymbol{\rho}_k \quad (9.4)$$

In other words, the matrix that contains the partial autocorrelation coefficients can be calculated from the inverse matrix of all the lagged autocorrelation coefficients multiplied by the vector that contains all the autocorrelations.

It sounds very complicated, but in terms of Excel functions this means that in order to calculate the partial autocorrelations, we need only two functions, =MMULT() and =MINVERSE(). The first function is used to multiply two matrices and the second one returns an inverse of the matrix, which is exactly what we need, as per equation (9.4).

There is an option to use another method to calculate the partial autocorrelation coefficients by using the determinants of the matrices via the Excel =MDETERM() function. However, this would not be any more elegant than via the =MMULT() and =MINVERSE() function.

The vector \mathbf{p}_k in the equation (9.4) shows that we first need to create a table with all the calculated autocorrelation coefficients. Fig. 9.1 shows just the first nine, although the full spreadsheet contains all twenty.

	A	B	C	D	E	F	G	H	I	J
1	1	0.80	0.66	0.58	0.51	0.39	0.28	0.12	0.04	-0.04
2	0.80	1	0.80	0.66	0.58	0.51	0.39	0.28	0.12	0.04
3	0.66	0.80	1	0.80	0.66	0.58	0.51	0.39	0.28	0.12
4	0.58	0.66	0.80	1	0.80	0.66	0.58	0.51	0.39	0.28
5	0.51	0.58	0.66	0.80	1	0.80	0.66	0.58	0.51	0.39
6	0.39	0.51	0.58	0.66	0.80	1	0.80	0.66	0.58	0.51
7	0.28	0.39	0.51	0.58	0.66	0.80	1	0.80	0.66	0.58
8	0.12	0.28	0.39	0.51	0.58	0.66	0.80	1	0.80	0.66
9	0.04	0.12	0.28	0.39	0.51	0.58	0.66	0.80	1	0.80
10	-0.04	0.04	0.12	0.28	0.39	0.51	0.58	0.66	0.80	1

Fig. 9.1 Autocorrelation coefficient matrix

Once this matrix has been created, we can start using the matrix functions in Excel that will enable us to calculate the partial autocorrelations. Fig. 9.2 shows the final result. The orange coloured cells are the full partial autocorrelation coefficients ϕ_{kk} , and the cells vertically above these coloured cells are the partial autocorrelation coefficients ϕ_{kj} , that are necessary for calculation (see a yellow box example after equation (8.3) for details).

	A	B	C	D	E	F	G	H	I	J	K
21		j									
22	k	1	2	3	4	5	6	7	8	9	10
23	1	0.800	0.747	0.742	0.738	0.746	0.734	0.716	0.727	0.729	0.733
24	2		0.066	0.010	0.009	0.016	0.028	0.000	-0.005	-0.018	-0.025
25	3			0.076	0.039	0.040	0.043	0.086	0.091	0.096	0.120
26	4				0.050	0.176	0.177	0.187	0.179	0.174	0.164
27	5					-0.170	-0.116	-0.110	-0.113	-0.104	-0.096
28	6						-0.072	0.108	0.108	0.112	0.097
29	7							-0.245	-0.275	-0.275	-0.283
30	8								0.042	0.077	0.079
31	9									-0.049	-0.111
32	10										0.085

Fig. 9.2 Partial autocorrelation coefficients for DJI time series

The cells are calculated as:

PACF for lag 1,1 Cell B23 Value (same as the first autocorrelation coefficient)

PACF for lag 2,1 Cells C23:C24 {=MMULT(MINVERSE(\$A\$1:B2),\$A\$2:\$A3)}

PACF for lag 2,2 Cells D23:D25 {=MMULT(MINVERSE(\$A\$1:C3),\$A\$2:\$A4)}

PACF for lag 3,3 Cells E23:E26 {=MMULT(MINVERSE(\$A\$1:D4),\$A\$2:\$A5)}

Compare the values in coloured cells in Fig. 9.2 (B23, C24, D25, E26, etc.) with the ones in column C (C3:C11) in Fig. 8.2 and you will see that they are identical.

Fig. 9.3 shows how the above formula has been implemented.

	A	B	C	D	E	F
21	j					
22	k	1	2	3	4	5
23	1	0.8	=MMULT(MINVERSE(\$A\$1:B2),\$A\$2:\$A3)	=MMULT(MINVERSE(\$A\$1:C3),\$A\$2:\$A4)	=MMULT(MINVERSE(\$A\$1:D4),\$A\$2:\$A5)	=MMULT(MINVERSE(\$A\$1:E5),\$A\$2:\$A6)
24	2		=MMULT(MINVERSE(\$A\$1:B2),\$A\$2:\$A3)	=MMULT(MINVERSE(\$A\$1:C3),\$A\$2:\$A4)	=MMULT(MINVERSE(\$A\$1:D4),\$A\$2:\$A5)	=MMULT(MINVERSE(\$A\$1:E5),\$A\$2:\$A6)
25	3			=MMULT(MINVERSE(\$A\$1:C3),\$A\$2:\$A4)	=MMULT(MINVERSE(\$A\$1:D4),\$A\$2:\$A5)	=MMULT(MINVERSE(\$A\$1:E5),\$A\$2:\$A6)
26	4				=MMULT(MINVERSE(\$A\$1:D4),\$A\$2:\$A5)	=MMULT(MINVERSE(\$A\$1:E5),\$A\$2:\$A6)
27	5					=MMULT(MINVERSE(\$A\$1:E5),\$A\$2:\$A6)

Fig. 9.3 Calculations for partial autocorrelation coefficients for DJI time series

The curly bracket in the formula $\{=MMULT(MINVERSE(\$A\$1:B2),\$A\$2:\$A3)\}$ are not entered manually. This is an array formula that is entered by clicking SHIFT, CTRL and ENTER at the same time after the formula has been entered into the formula bar. The curly brackets are added automatically by Excel.

To execute this formula, we need to follow these steps:

1. We first highlight the cells where the results will be displayed (see in Fig 9.2 an example of the range F23:F27)
2. While the cells are highlighted, go to the formula bar and enter the formula: $=MMULT(MINVERSE(\$A\$1:E5),\$A\$2:\$A6)$. Do not press ENTER!
3. Press SHIFT, CTRL and ENTER together
4. The values get populated in the range F23:F27 in Fig. 9.2

In our formula above, the first part represents the inverse of the cells A1:E5 (which is equivalent to P_k^{-1} in equation 9.4) and A2:A6 is the vector (equivalent to ρ_k in equation (9.4)). Fig 9.4 illustrates the ranges that the formula covers.

	A	B	C	D	E	F	G	H
1	1	0.80	0.66	0.58	0.51	0.39	0.28	0.12
2	0.80	1	0.80	0.66	0.58	0.51	0.39	0.28
3	0.66	0.80	1	0.80	0.66	0.58	0.51	0.39
4	0.58	0.66	0.80	1	0.80	0.66	0.58	0.51
5	0.51	0.58	0.66	0.80	1	0.80	0.66	0.58
6	0.39	0.51	0.58	0.66	0.80	1	0.80	0.66
7	0.28	0.39	0.51	0.58	0.66	0.80	1	0.80
8	0.12	0.28	0.39	0.51	0.58	0.66	0.80	1

Fig. 9.4 How matrix functions are used to calculate the partial autocorrelation coefficients

It might appear complicated to start with, but this is a reasonably simple way to calculate the partial autocorrelations in Excel. The value of this method is that it actually shows you what is happening. However, we can automate this even more and bring the calculations of the partial autocorrelation coefficients to a single formula.

Fig. 9.5 below is identical to Fig. 9.1, except that we added three new columns. Column S contains just sequential numbers, representing the lags for the partial autocorrelation coefficients. Columns T and V have identical values, and they are the partial autocorrelation coefficients, but calculated using a single formula in two different ways. Let's explain.

	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O	P	Q	R	S	T	V
1	1	0.80	0.66	0.58	0.51	0.39	0.28	0.12	0.04	-0.04	-0.07	-0.12	-0.11	-0.10	-0.09	-0.06	-0.01	-0.01	1	0.800	0.800
2	0.80	1	0.80	0.66	0.58	0.51	0.39	0.28	0.12	0.04	-0.04	-0.07	-0.12	-0.11	-0.10	-0.09	-0.06	-0.01	2	0.066	0.066
3	0.66	0.80	1	0.80	0.66	0.58	0.51	0.39	0.28	0.12	0.04	-0.04	-0.07	-0.12	-0.11	-0.10	-0.09	-0.06	3	0.076	0.076
4	0.58	0.66	0.80	1	0.80	0.66	0.58	0.51	0.39	0.28	0.12	0.04	-0.04	-0.07	-0.12	-0.11	-0.10	-0.09	4	0.050	0.050
5	0.51	0.58	0.66	0.80	1	0.80	0.66	0.58	0.51	0.39	0.28	0.12	0.04	-0.04	-0.07	-0.12	-0.11	-0.10	5	-0.170	-0.170
6	0.39	0.51	0.58	0.66	0.80	1	0.80	0.66	0.58	0.51	0.39	0.28	0.12	0.04	-0.04	-0.07	-0.12	-0.11	6	-0.072	-0.072
7	0.28	0.39	0.51	0.58	0.66	0.80	1	0.80	0.66	0.58	0.51	0.39	0.28	0.12	0.04	-0.04	-0.07	-0.12	7	-0.245	-0.245
8	0.12	0.28	0.39	0.51	0.58	0.66	0.80	1	0.80	0.66	0.58	0.51	0.39	0.28	0.12	0.04	-0.04	-0.07	8	0.042	0.042
9	0.04	0.12	0.28	0.39	0.51	0.58	0.66	0.80	1	0.80	0.66	0.58	0.51	0.39	0.28	0.12	0.04	-0.04	9	-0.049	-0.049
10	-0.04	0.04	0.12	0.28	0.39	0.51	0.58	0.66	0.80	1	0.80	0.66	0.58	0.51	0.39	0.28	0.12	0.04	10	0.085	0.085
11	-0.07	-0.04	0.04	0.12	0.28	0.39	0.51	0.58	0.66	0.80	1	0.80	0.66	0.58	0.51	0.39	0.28	0.12	11	0.012	0.012
12	-0.12	-0.07	-0.04	0.04	0.12	0.28	0.39	0.51	0.58	0.66	0.80	1	0.80	0.66	0.58	0.51	0.39	0.28	12	0.105	0.105
13	-0.11	-0.12	-0.07	-0.04	0.04	0.12	0.28	0.39	0.51	0.58	0.66	0.80	1	0.80	0.66	0.58	0.51	0.39	13	0.016	0.016
14	-0.10	-0.11	-0.12	-0.07	-0.04	0.04	0.12	0.28	0.39	0.51	0.58	0.66	0.80	1	0.80	0.66	0.58	0.51	14	-0.026	-0.026
15	-0.09	-0.10	-0.11	-0.12	-0.07	-0.04	0.04	0.12	0.28	0.39	0.51	0.58	0.66	0.80	1	0.80	0.66	0.58	15	0.035	0.035
16	-0.06	-0.09	-0.10	-0.11	-0.12	-0.07	-0.04	0.04	0.12	0.28	0.39	0.51	0.58	0.66	0.80	1	0.80	0.66	16	0.013	0.013
17	-0.01	-0.06	-0.09	-0.10	-0.11	-0.12	-0.07	-0.04	0.04	0.12	0.28	0.39	0.51	0.58	0.66	0.80	1	0.80	17	-0.063	-0.063
18	-0.01	-0.01	-0.06	-0.09	-0.10	-0.11	-0.12	-0.07	-0.04	0.04	0.12	0.28	0.39	0.51	0.58	0.66	0.80	1	18	0.051	0.051
19	0.03																				

Fig. 9.1 Calculating the partial autocorrelations (PACF) with a single Excel formula

Excel formula for inverse matrix, as we just learned, is: =MINVERSE(array). The red font refers to an array that represents the original matrix from which an inverse matrix will be created. So, if we say =MINVERSE(\$A\$1:C3), this means that the original matrix is in cells A1:C3.

Excel formula for multiplying two matrices, as we also learned, is: =MMULT(array1, array2). The red and green are the two matrices that will be multiplied. So, if we say =MMULT(MINVERSE(\$A\$1:C3), \$A\$2:\$A4), we are multiplying an inverse of the square matrix (\$A\$1:C3) with a single column matrix (\$A\$2:\$A4). As we know the result will be a single column matrix.

And finally, we'll also use Excel formula for index: =INDEX(array, row_num, [column_num]), or the other form, which is: =INDEX(reference, row_num, [column_num], [area_num]). The segments in square brackets are optional. Let's break down this formula using an example.

If, for example, we say =INDEX(A5:C10,3,2), the first part A5:C10 is an array from which we are going to extract the value. To extract the value, we find the intersection between the third row in this area and the second column. The result is the value that is in cell C8. Why? The first cell in this example array is A5, so we go 3 rows down (which is to row 8). Again, we start with A5 and go two columns to the right, which is the column C. So, the intersection between row 8 and column C is the cell C8. Therefore, in response to the function =INDEX(), Excel returns the value from this cell.

We can combine all the previous formulas into one single formula that will give us the partial autocorrelation coefficients. For example:
=INDEX(MMULT(MINVERSE(\$A\$1:C3), \$A\$2:\$A4), COUNT(\$A\$2:\$A4)).

What we are doing here is exactly what we did before (see Fig. 5.6.4), which is: multiplying the matrix A2:A4 with the inverse matrix of A1:C3. The =INDEX() function is saying that from the result of this multiplication (which will be a column of numbers), we want just the last value in the column. The part that says COUNT(\$A\$2:\$A4) gives us this. It counts how many elements are in this column and only the last one is displayed.

To see how this was applied, look at the formulae in column T. You can see the following:

T1=INDEX(MMULT(MINVERSE(\$A\$1:A1),\$A\$2:\$A2),COUNT(\$A\$2:\$A2))
 T2=INDEX(MMULT(MINVERSE(\$A\$1:B2),\$A\$2:\$A3),COUNT(\$A\$2:\$A3))
 T3=INDEX(MMULT(MINVERSE(\$A\$1:C3),\$A\$2:\$A4),COUNT(\$A\$2:\$A4))
 .
 .
 T17 =INDEX(MMULT(MINVERSE(\$A\$1:Q17),\$A\$2:\$A18),COUNT(\$A\$2:\$A18))

So, column T gives us the partial autocorrelation coefficients. This is exactly the same set of formulae that we used in our previous example that helped us create the partial autocorrelation coefficients in the range B23:R39 (Fig. 5.6.2). The only difference is that we do not see all the intermediate coefficients $r_{k,j}$, but only the final ones $r_{k,k}$, which is what we want. The only problem with the formulae here are the red marked parts. If you copy the formula down, it will copy everything correctly, except the red parts. You will need to manually change A1, to B2, to C3, etc.

A solution to this problem is to make the matrices relative, rather than use the explicit ranges. To do this, we need Excel formula =OFFSET(reference, rows, cols, [height], [width]). The last two elements in square brackets are optional. Let's use an example to explain this function.

If we say =OFFSET(D3,3,-2,1,1), this means that the starting point is the cell D3, which is the reference cell. From there, we go three rows down (which is to row 6) and then move 2 columns to the left (which is to column B). The result is B6. In response to the function =OFFSET(), Excel returns the value from this cell.

We can now use this format to create a one single formula as follow:

=INDEX(MMULT(MINVERSE(OFFSET(\$A\$1,0,0,S3,S3)),OFFSET(\$A\$1,1,0,S3,1)),S3).

What we are saying is that we first multiply the matrix that is defined by the OFFSET function as (\$A\$1,1,0,S3,1), which is the same as (\$A\$2:\$A4), with an inverse matrix in (\$A\$1,0,0,S3,S3), which is the same as \$A\$1:C3. The result will be a column of numbers, but we want the last one which is actually the last value in column S, so in our case the value from cell S3. Instead of S3 in the above formula, we could have used COUNT(\$A\$2:\$A4). Either way it gives us the same number for the number of lags.

To see how this was applied, look at the formulae in column V. You can see the following:

V1=INDEX(MMULT(MINVERSE(OFFSET(\$A\$1,0,0,S1,S1)),OFFSET(\$A\$1,1,0,S1,1)),S1)
 V2=INDEX(MMULT(MINVERSE(OFFSET(\$A\$1,0,0,S2,S2)),OFFSET(\$A\$1,1,0,S2,1)),S2)
 V3=INDEX(MMULT(MINVERSE(OFFSET(\$A\$1,0,0,S3,S3)),OFFSET(\$A\$1,1,0,S3,1)),S3)
 .
 .
 V17=INDEX(MMULT(MINVERSE(OFFSET(\$A\$1,0,0,S17,S17)),OFFSET(\$A\$1,1,0,S17,1)),
 S17)

Now it is clear that if you copied cell V1 down, there will be no need for manual corrections. This is now fully automated formula for calculating the partial autocorrelation coefficients with a single line and a Copy/Paste action.

14. Fitting the model

Now we obtained preliminary estimates of the model coefficients, we can fit it to the actual data.

Fitting and forecasting can be quite confusing and in order to be explicit, we are summarising a selection of the most basic formulae used for various Box-Jenkins models. It is very important to remember that these formulae imply the following relationships:

For $d=0$: $y_t = y_t - \mu$, and as $z_t = y_t$ and $\mu = \bar{y}$, this means that $z_t = y_t - \bar{y}$

For $d=1$: $z_t = y_t - y_{t-1}$

For $d=2$: $z_t = \nabla^2 z_t = z_t - z_{t-1} = y_t - y_{t-1} - y_{t-1} - y_{t-2} = y_t - 2y_{t-1} + y_{t-2}$

For $d=3$: $z_t = \nabla^3 z_t = \nabla^2 z_t - \nabla^2 z_{t-1} = y_t - 3y_{t-1} + 3y_{t-2} - y_{t-3}$, etc.

Also remember that: $\hat{y}_t(1) = \hat{y}_{t+1}$. In other words, the forecast for just one period ahead at the time t , is in fact an estimated value of that one period ahead.

To decide if the constant δ is needed, and what is the value of the constant, Table 14.1 below is put together to help us with the decision-making process.

	AR only	MA only
Stationary ($d=0$)		
Zero mean	δ not needed	δ not needed
Non-zero mean	$\delta = \bar{y}_t (1 - \phi_1)$ for AR(1)	$\delta = \bar{y}$
	$\delta = \bar{y}_t (1 - \phi_1 - \phi_2)$ for AR(2)	$\delta = \bar{y}$
	$\delta = \bar{y}_t (1 - \phi_1 - \phi_2 - \phi_3)$ for AR(3), etc.	$\delta = \bar{y}$
Nonstationary ($d \geq 1$)	δ not needed	δ not needed

Table 14.1 Decision criteria for the inclusion of δ in the model

And finally, remember that:

$$a_t = y_t - \hat{y}_{t-1}(1), \text{ or } a_t = y_t - \hat{y}_t$$

$$a_{t-1} = y_{t-1} - \hat{y}_{t-2}(1), \text{ or } a_{t-1} = y_{t-1} - \hat{y}_{t-1}, \text{ etc.}$$

Earlier we gave example of several AR(p), MA(q) and ARMA(p,q) models. To refresh your memory, here they are again:

AR(1) $z_t = \phi_1 z_{t-1} + a_t$

AR(2) $z_t = \phi_1 z_{t-1} + \phi_2 z_{t-2} + a_t$

AR(3) $z_t = \phi_1 z_{t-1} + \phi_2 z_{t-2} + \phi_3 z_{t-3} + a_t$ etc.

MA(1) $z_t = -\theta_1 a_{t-1} + a_t$

MA(2) $z_t = -\theta_1 a_{t-1} - \theta_2 a_{t-2} + a_t$

MA(3) $z_t = -\theta_1 a_{t-1} - \theta_2 a_{t-2} - \theta_3 a_{t-3} + a_t$ etc.

ARMA(1,1) $z_t = \phi_1 z_{t-1} - \theta_1 a_{t-1} + a_t$

ARMA(2,2) $z_t = \phi_1 z_{t-1} + \phi_2 z_{t-2} - \theta_1 a_{t-1} - \theta_2 a_{t-2} + a_t$

ARMA(3,3) $z_t = \phi_1 z_{t-1} + \phi_2 z_{t-2} + \phi_3 z_{t-3} - \theta_1 a_{t-1} - \theta_2 a_{t-2} - \theta_3 a_{t-3} + a_t$ etc.

Given the above summary of how z_t is created and if δ is necessary, the same formulae can be expressed in terms of y_t :

$$\begin{aligned} \text{AR(1)} \quad & y_t = \mu(1 - \phi_1) + \phi_1 y_{t-1} + a_t \\ \text{AR(2)} \quad & y_t = \mu(1 - \phi_1 - \phi_2) + \phi_1 y_{t-1} + \phi_2 y_{t-2} + a_t \\ \text{AR(3)} \quad & y_t = \mu(1 - \phi_1 - \phi_2 - \phi_3) + \phi_1 y_{t-1} + \phi_2 y_{t-2} + \phi_3 y_{t-3} + a_t \text{ etc.} \\ \\ \text{MA(1)} \quad & y_t = \mu - \theta_1 a_{t-1} + a_t \\ \text{MA(2)} \quad & y_t = \mu - \theta_1 a_{t-1} - \theta_2 a_{t-2} + a_t \\ \text{MA(3)} \quad & y_t = \mu - \theta_1 a_{t-1} - \theta_2 a_{t-2} - \theta_3 a_{t-3} + a_t \text{ etc.} \\ \\ \text{ARMA(1,1)} \quad & y_t = \mu(1 - \phi_1) + \phi_1 y_{t-1} - \theta_1 a_{t-1} + a_t \\ \text{ARMA(2,2)} \quad & y_t = \mu(1 - \phi_1 - \phi_2) + \phi_1 y_{t-1} + \phi_2 y_{t-2} - \theta_1 a_{t-1} - \theta_2 a_{t-2} + a_t \\ \text{ARMA(3,3)} \quad & y_t = \mu(1 - \phi_1 - \phi_2 - \phi_3) + \phi_1 y_{t-1} + \phi_2 y_{t-2} + \phi_3 y_{t-3} - \theta_1 a_{t-1} - \theta_2 a_{t-2} - \theta_3 a_{t-3} + a_t \text{ etc.} \end{aligned}$$

On the basis of what we estimated so far, i.e. $\phi_1=0.72$ and $\theta_1=0.0064$, and given the above formulae for ARMA(1,1), our model expressed in terms of z_t is:

$$z_t = 0.72z_{t-1} - 0.0064a_{t-1} + a_t$$

Armed with this, we can go back to our ARMA(1,1) model from the previous chapter and show how to fit the model in terms of z_t and y_t . As before, we'll rely on the templates from the spreadsheets. The template is shown in Fig. 14.1. This is a copy of the sheet called "5. Model" from the workbook.

	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O	P
1	Time Series z_t	Value	Comment		Fitted Series \hat{z}_t	Value	Comment			Period	y_t	$z_t - y_t - \mu$	AR(p)	MA(q)	\hat{z}_t	e_t
2	Number of observations	50			Fitted Series \hat{z} Mean μ	1.68				0						
3	Mean μ	1.78			Fitted Series \hat{z} St. Dev. σ	2.69				1	1.30	-0.48	0.00	0.00	0.00	0.00
4	St. Dev. σ	3.79			R-Squared	0.52				2	8.60	6.82	0.93	0.00	1.44	7.16
5	Criterion for use of δ	1.05	Use Delta from B6							3	12.13	10.35	6.16	0.05	6.62	5.51
6	δ	0.51								4	10.97	9.19	8.69	0.04	9.16	1.81
7	95% Confid. 1.96*SE	1.05	Non-zero mean							5	7.85	6.08	7.85	0.01	8.35	-0.49
8	Implied μ	1.78								6	6.93	5.15	5.62	0.00	6.13	0.80
9										7	3.87	2.09	4.96	0.01	5.46	-1.60
10	Permissible Region	Value	Comment							8	2.94	1.16	2.77	-0.01	3.28	-0.34
11	AR(1) $-1 < \phi_1 < 1$	0.72	OK		Copy to cells F7:G9 the appropriate values of ϕ and θ from the cells in column B in Sheet "2. Phi!" for ϕ and column B in Sheet "3. Covar!" for θ . Important: Put in cells F6:G6 the number of p and q parameters from from your selected model. Otherwise, the fitting will not work.					9	-0.12	-1.90	2.11	0.00	2.62	-2.74
12	MA(1) $-1 < \theta_1 < 1$	0.01	OK							10	4.95	3.17	-0.09	-0.02	0.43	4.52
13	AR(2) $-1 < \phi_2 < 1$	0.00	OK							11	3.28	1.50	3.54	0.03	4.02	-0.74
14	AR(2) $\phi_1 + \phi_2 < 1$	0.72	OK							12	-0.09	-1.87	2.95	0.00	2.86	-2.95
15	AR(2) $\phi_2 - \phi_1 < 1$	-0.72	OK							13	2.86	1.09	-0.06	-0.02	0.46	2.40
16	MA(2) $-1 < \theta_2 < 1$	0.00	OK							14	3.35	1.57	2.05	0.02	2.54	0.80
17	MA(2) $\theta_1 + \theta_2 < 1$	0.01	OK							15	1.84	0.06	2.40	0.01	2.90	-1.06
18	MA(2) $\theta_2 - \theta_1 < 1$	-0.01	OK							16	4.54	2.76	1.32	-0.01	1.83	2.71
19	AR(3) $-1 < \phi_3 < 1$	0.00	OK							17	0.91	-0.87	3.25	0.02	3.74	-2.82
20	AR(3) $\phi_1 + \phi_2 + \phi_3 < 1$	0.72	OK							18	1.77	-0.01	0.65	-0.02	1.18	0.59
21	AR(3) $-\phi_1 + \phi_2 - \phi_3 < 1$	-0.72	OK							19	2.41	0.63	1.26	0.00	1.77	0.64
22	AR(3) $\phi_3(\phi_2 - \phi_1) - \phi_2 < 1$	0.00	OK							20	4.96	3.18	1.72	0.00	2.22	2.74
23	MA(3) $-1 < \theta_3 < 1$	0.00	OK							21	1.29	-0.49	3.55	0.02	4.04	-2.76
24	MA(3) $\theta_1 + \theta_2 + \theta_3 < 1$	0.01	OK							22	2.11	0.33	0.92	-0.02	1.44	0.67
25	MA(3) $-\theta_1 + \theta_2 - \theta_3 < 1$	-0.01	OK							23	2.75	0.97	1.51	0.00	2.01	0.73
26	MA(3) $\theta_3(\theta_2 - \theta_1) - \theta_2 < 1$	0.00	OK							24	1.30	-0.48	1.97	0.00	2.47	-1.17

Fig. 14.1 Template for fitting the time series with the estimated model

The template has several coloured boxes to help with the calculations and data interpretation. The actual time series is copied from previous sheets to column K, whilst the column L again contains the differences of every observation from the overall mean. Columns M and N calculate the appropriate AR and MA component, whilst column O is the combination of the

two, i.e. one step forecasts, or a fit, for our time series. Column P contains the errors generated by fitting the forecasts (model data) to the actual time series.

The blue coloured box A1:C7 contains the most elementary descriptive statistics for the time series, including the rule whether to include δ in the equation. The green box E1:G4 contains a few details of the fitted series (which we will do shortly) and enable a comparison between the fitted and actual dataset. The two yellow boxes contain all the details that are related to the coefficients. The box E6:H9 contains the preliminary estimated values of the coefficients. We can see that cells F9 and G9 contains the values we already estimated for ϕ_1 , and θ_1 . The other two cells for ϕ_2 , ϕ_3 , θ_2 and θ_3 are empty as we did not have them in our ARMA(1,1) model.

It is very **important to emphasise** that you must insert the right number of parameters in the cells F6 and G6 from Fig. 14.1 (in our case 1 and 1). If you do not do this, the forecasts in the template will not be executed correctly.

By looking in the cells in columns M and N you will see some IF statements that pick the correct forecasting formula depending on what numbers are placed in cells F6 and G6. Also, if you are using a different model, the correct formula needs to be created in the cell B4 for δ . For details see Chapter 10, as well as Table 14.1.

As we said above, the values in columns M and N calculate the AR and the MA part of the model. The column O, that contains \hat{z}_t values, brings it all together by adding the values from the cells in M and N, plus the value of the constant δ from B6. And finally, the column P contains the forecasting errors e_t .

Our model, expressed in terms of y_t is:

$$y_t = 0.51(1 - 0.72) + 0.72y_{t-1} - 0.0064_1 a_{t-1} + a_t = 0.1428 + 0.72y_{t-1} - 0.0064_1 a_{t-1} + a_t$$

How well our model fits the actual data, can be seen in Fig. 14.2.




Fig. 14.2 The actual data and the model ARMA(1,1) fit

Before we either enthusiastically embrace the model, or discard it, we need to subject it to some diagnostic validation. This will tell us if the model is acceptable at all. In fact, if we were unsure and have used several tentative models, during this next stage of diagnostic checking we will be in the position to decide which one is the most acceptable.

18. Alternative shortcuts to estimation and forecasting

What we described so far covers the “textbook case” of how to estimate the parameters. However, if we are just doing some preliminary work and experimenting with data, the procedure might appear to be quite lengthy. A quicker method of estimating the model coefficients would be desirable. As it happens, using Excel Solver function we might be able to achieve this. We created two templates that will enable us go through this phase at greater speed and elegance.

The **first template** (see workbook “BJ Chapter 18A.xlsx”) utilizes the formulae that Box and Jenkins suggested to be used for initial estimates of the parameters ϕ and θ . This is all done in one simple sheet (see the sheet “2. Estim” in the above workbook). However, what precedes this sheet is the sheet containing ACF and PACF calculations (because we need to decide what category of models our time series belongs), just as in the previous workbook. The sheets that follow are the modelling sheets and the forecasting sheet, as before. We have in this case omitted the residual analysis and everything else that, in fact, has to be done to validate the model. However, we know that this has to be done.

Fig. 18.1 contains the details of how to calculate the preliminary estimates for ϕ and θ . These formulae were already shown in Table 11.1.

	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O
1	ACF y_t	PACF y_t		Solve	ϕ	θ		Model	Formulae			Value	Solution	Conditions	
2	0.708083	0.708083		E2=A2		0.71		AR(1)	$r_1 = \phi_1$			0.708083	OK	$ \phi_1 < 1$	
3	0.506153	0.009568													
4	0.422608	0.122077													
5	0.218601	-0.25499		E5=(A2*(1-A3))/(1-A2^2)		0.70		AR(2)	$\phi_1 = \frac{r_1(1-r_2)}{1-r_2^2}$			0.710876	OK	$\phi_2 + \phi_1 < 1$	
6	0.145197	0.117752		E6=(A3-A2^2)/(1-A2^2)		0.01			$\phi_2 = \frac{r_2-r_1^2}{1-r_1^2}$			-0.69174	OK	$\phi_2 - \phi_1 < 1$	
7	0.144989	0.045343										0.009568	OK	$ \phi_2 < 1$	
8	0.00724	-0.1843													
9	0.019183	0.151204													
10	-0.01045	-0.16056		D10=((F10)/(1+F10^2))-A2	-0.2080834		-1.00	MA(1)	$r_1 = \frac{-\theta_1}{1+\theta_1^2}$			1.000001	No	$ \theta_1 < 1$	
11	-0.07656	0.061712													
12	-0.01344	0.05763													
13	0.028666	0.049159		D13=(((-F13*(1-F14))/(1+F13^2+F14^2))-A2	-0.03726301		-1.12	MA(2)	$r_1 = \frac{-\theta_1(1-\theta_2)}{1+\theta_1^2+\theta_2^2}$			-1.61803	OK	$\theta_2 + \theta_1 < 1$	
14	-0.00899	-0.04796		D14=(((-F13)/(1+F13^2+F14^2))-A3	-0.05893907		-0.50		$r_2 = \frac{-\theta_1}{1+\theta_1^2+\theta_2^2}$			0.618034	OK	$\theta_2 - \theta_1 < 1$	
15	0.01392	-0.05372		D15=SUM(D13:D14)	-0.09620208							0.5	OK	$ \theta_2 < 1$	
16	0.030127	0.104425													
17	0.059035	0.066177													
18	0.097326	-0.00261		D18=((1-E18*F18)*(E18-F18))/(1+F18^2-2*E18*F18))-A2	0.060437571	0.63	-0.37	ARMA(1,1)	$r_1 = \frac{(1-\phi_1\theta_1)(\phi_1-\theta_1)}{1+\theta_1^2-2\phi_1\theta_1}$			0.629468	OK	$ \phi_1 < 1$	
19	0.08894	0.0212		D19=(E18*A2)-A3	-0.06043712							0.368364	OK	$ \theta_1 < 1$	
20	-0.01103	-0.0145		D20=SUM(D18:D19)	4.49398E-07				$r_2 = \phi_1 r_1$						

Fig. 18.1 Calculation of the preliminary estimates for ϕ and θ

Columns A and B contain the values of ACF and PACF, as per previous examples and templates. Columns E and F (or to be precise, the cells E2, E5 & E6, F10, F13 & F14 and E18 & F18) contain the values of the preliminary estimates of ϕ and θ . Unless there is an explicit formula, such as E5 or E6, the estimates were achieved by utilizing the Solver. Let’s look at the example for ARMA(1,1) model, where we are trying to estimate E18 and F18 for ϕ_1 and θ_1 . Fig. 18.2 shows an example how the Solver dialogue box was structured.

As before, we declared D20 to be the objective with the value of zero. We are going to meet this objective by changing the cells E18:F18, with the constraints that certain conditions, as per cells L18:L19 are met. So far, so good. However, the key to this optimization is in cells D18:D20. In fact cells C18:C20 show the Excel formulae describing what is going on in D18:D20. They are just translations of the actual formulae that you can also see in the sheet. Fundamentally, we are saying that D18 and D19, as well as their sum in D20 need to add to zero (this is the Objective in the Solver dialogue box).

As D20 turns out to be 4.49398E-07 (Excel speak for 0.000000449398), which is virtually zero, the Solver has automatically inserted the values of ϕ_1 and θ_1 into cells E18 and F18. They are 0.63 and -0.37 for ϕ_1 and θ_1 respectively. Compare this with the values for ϕ_1 and θ_1 from Chapter 13, where they were estimated as 0.72 and 0.0064. If you follow the example, you will see that the AIC, BIC and MSE are marginally worse than those from Chapter 13. Another red flag are the negative serial correlation as per cell F30. The coefficients need further tuning.




Fig. 18.2 Solver dialogue box for preliminary estimates of ϕ and θ

The **second template** (see workbook “BJ Chapter 18B.xlsx”) is taking it to the next level of simplicity, where we literary let Solver pick the model. It is a crude approximation, but for the purposes of quick execution, it has some value. In order to see how does this template work, let’s check Fig. 18.3.

Time Series	Value	Comment	Fitted Series \hat{z}_t	Value	Comment	y_t	AR(p)	MA(q)	\hat{y}_t	e_t
Number of observations	50		Fitted Series \hat{z}_t Mean μ	2.70		1.30	0.00	0.00	0.00	0.00
Mean μ	1.78		Fitted Series \hat{z}_t St. Dev. σ	2.59		8.60	1.04	0.00	2.09	6.51
St. Dev. σ	3.79		R-Squared	0.47		12.13	6.88	1.64	6.29	5.85
Criterion for use of δ	1.05	Use Delta from B6				10.97	9.70	1.47	9.28	1.69
δ	0.36					7.85	8.77	0.43	9.40	-1.54
95% Confid. 1.96*SE	0.10	Non-zero mean				6.93	6.28	-0.39	7.72	-0.79
Implied μ	1.78					3.87	5.54	-0.20	6.79	-2.93
						2.94	3.09	-0.74	4.88	-1.94
						-0.12	2.36	-0.49	3.89	-4.02
						4.95	-0.10	-1.01	1.96	2.99
						3.28	3.96	0.75	4.26	-0.98
						-0.09	2.62	-0.25	3.92	-4.01
						2.86	-0.07	-1.01	1.99	0.87
						3.35	2.29	0.22	3.12	0.23
						1.84	2.68	0.06	3.67	-1.83
						4.54	1.47	-0.46	2.98	1.55
						0.91	3.63	0.39	4.29	-3.37
						1.77	0.73	-0.85	2.63	-0.87
						2.41	1.41	-0.22	2.68	-0.27
						4.96	1.92	-0.07	3.04	1.92
						1.29	3.97	0.48	4.54	-3.25
						2.11	1.03	-0.82	2.90	-0.78
						2.75	1.69	-0.20	2.94	-0.19
						1.30	2.20	-0.05	3.29	-2.00
						0.02	1.04	-0.50	2.59	-2.57
						0.81	0.02	-0.65	1.71	-0.91
						-0.58	0.65	-0.23	1.92	-2.50
						-1.83	-0.46	-0.63	1.22	-3.05
						-3.03	-1.46	-0.77	0.35	-3.39
						-6.21	-2.43	-0.85	-0.52	-5.69

Fig. 18.3 Initial template for estimating ARIMA(p,d,q) coefficients ϕ and θ

This workbook only contains only two sheets, “1. ACF_PACF” and “2. B_J ARMA“. The template looks very similar to all the previous ones, and indeed the majority of the formulae have been kept unchanged. The only difference is how we invoke the Solver function.

We will again invoke the Solver function, but before that we just inserted the values of 0.001 to all cells between F7:G9. Effectively these are our initial seed values for all ϕ and θ coefficients up to the model of $p=3$ and $q=3$. In our case, because we are using the same time series and we know that $d=0$, so what remain to be estimated are the values for up to ϕ_3 and θ_3 .

To start the process, we will invoke the Solver function and the dialogue box is displayed in Fig. 18.4.




Fig. 18.4 Solver dialogue box for estimating all ϕ and θ coefficients in an ARMA(p,q) model

As you can see, our Objective is to minimize the AIC value, which is in cell F21. We are going to change cells F6:G6 between 1 and 3, each cell. This means that we will take all the combinations between ϕ_1 to ϕ_3 and θ_1 to θ_3 , and seek their values in cells F7:G9, subject to constraints as shown.

Once the Solver completed the iterations, the model suggested was ARMA(1,1) with $\phi_1 = 0.80$ and $\theta_1 = 0.25$. As we can see, it is the same model we selected and the values of the parameters are more or less in the same region. However, unlike the model from the previous template, this one does not have negatively correlated residuals.

The advantage of both templates is that we can treat these estimated coefficient values as just the preliminary estimates, which means we can improve on them. We'll let you experiment with the Excel Solver function to figure out how to do it.